


esempio n.1

* squadra RC (passa basso)


```
V1 1 0 AC 1
R1 1 2 1000
C1 2 0 100e-9
.AC DEC 10 1 100000
.PRINT AC V(2)
.END
```



esempio n.2


* analisi in CC

```
V1 1 0 DC 2 ; gen. di tensione continua (2V)
R1 1 2 1k
L1 2 3 100m
R2 3 0 1k
C1 3 0 220n
.END
```



esempio n.3



```
V1 1 0 PWL 0 0 1n 2 10 2 ; gradino di tensione
R1 1 2 1k
L1 2 3 100m
R2 3 0 1k
C1 3 0 220n
.TRAN 1n 2m
.PRINT TRAN V(3) V(1)-V(2)
.END
```



esempio n.4 - ampl. invertente


*con A.O. ideale

```
V1 1 0 AC 1
R1 1 2 1k
R2 2 3 100k
X1 0 2 3 0 AOID
.AC DEC 10 1 100k
.PLOT AC V(3)
*-----
.SUBCKT AOID 1 2 3 4
* nodi:      in+  in-  out  gnd
RIN 1 2 1e12
E1 3 4 1 2 1e9
.ENDS AOID
*-----
.END
```




esempio n.5 - ampl. invertente
 *con A.O. "reale" (risp. in frequenza)

```
V1 1 0 AC 1
R1 1 2 1k
R2 2 3 100k
X1 0 2 3 0 AORE
.AC DEC 10 1 100k
.PLOT AC V(3)
```




```
*-----
.SUBCKT AORE 1 2 3 4
* nodi:      Vin+  Vin-  Vo  massa
RI 1 2 2MEG
RO 6 3 75
GB 4 5 1 2 1M
R1 5 4 1K
C1 5 4 15.9u
EA 6 4 5 4 1E5
.ENDS AORE
*-----
.END
```



esempio n.6 - ampl. operazionale "reale"
 *(risp. ad anello aperto)


```
V1 1 0 AC 1
X1 1 0 2 0 AORE
.AC DEC 10 1 10k
.PLOT AC V(2)
```



```
*-----
.SUBCKT AORE 1 2 3 4
* nodi:      Vin+  Vin-  Vo  massa
RI 1 2 2MEG
RO 6 3 75
GB 4 5 1 2 1M
R1 5 4 1K
C1 5 4 15.9u
EA 6 4 5 4 1E5
.ENDS AORE
*-----
.END
```

esempio n.7
 *caratteristiche d'uscita di un BJT

```
* DC sweep
Q1 2 1 0 Q2N2222
I1 0 1 DC 0
V1 3 0 DC 0
R1 3 2 1
.DC V1 0 12 .1 I1 2u 10u 2u
.MODEL Q2N2222 NPN IS=14.3E-15 BF=256 VAF=74.03
.PRINT DC I(V1)
*.OPTION (LIMPTS=50000)
.END
```



esempio n.8

*punto di riposo di un BJT

Q1 2 4 3 Q2N2222

V1 1 0 DC 12

R1 1 4 22k

R2 4 0 6.8k

RC 1 2 2.2k


RE 3 0 1k

.OP

.MODEL Q2N2222 NPN IS=14.3E-15 BF=256

+VAF=74.03

.END



esempio n.9

*amplificatore a BJT

Q1 2 4 3 Q2N2222

V1 1 0 DC 12

R1 1 4 22k

R2 4 0 6.8k

RC 1 2 2.2k

RE 3 0 1k

*-----

V2 6 0 AC 1

RS 6 5 470

CS 5 4 22u

CE 3 0 4.7u

.AC DEC 10 1 100k

.MODEL Q2N2222 NPN IS=14.3E-15 BF=256

+VAF=74.03

.PRINT AC V(2)

.END

